

# Mikrocontroller-Grundlagen, Teil 11

Im elften Teil dieser Artikelserie fahren wir mit der Beschreibung der unterschiedlichen Timer-Betriebsarten fort.

## 4.1 Timer-Mode 0

Abbildung 96 zeigt die Funktionsweise des Timer/Zählers 0 (1) in der Betriebsart 0. Das Steuerbit C/T (Counter/Timer) nimmt die Auswahl zwischen der Timer- und Zählerfunktion vor. In der Timerfunktion (C/T = 0) erhält der nachgeschaltete Zähler direkt den durch 12 geteilten Oszillatortakt.

Die Zählung am Eingang des 5-Bit-Zählers ist dann freigegeben, wenn das Timer-Run-Bit TR 0 (1) gesetzt ist und außerdem entweder das Steuerbit Gate auf Low- oder der Eingangspin INT 0 (1) auf High-Pegel gesetzt ist. Bei gesetztem Gate-Steuersbit kann somit der nachgeschaltete Zähler über den externen Eingang INT 0 (1) gesteuert werden, wodurch beispielsweise eine Pulsbreitenmessung erleichtert wird.

Mit jeder fallenden Flanke am Ausgang der Steuerungseinheit erhöht sich der Inhalt des 5-Bit-Zählers um 1. TL 0 (1) verhält sich somit wie ein Vorteiler durch 32, während TH 0 (1) den Eingangstakt durch maximal 256 teilen kann. Dessen Ausgang kann zusätzlich beim Timer 1 für die Schiebepaketzeugung der seriellen Schnittstelle dienen, worauf wir im Verlauf der Artikelserie noch näher eingehen.

Mit jedem Überlauf der Timer-Kombination, d. h. beim Übergang vom Zustand, in dem alle Bits des Zählers 1 sind, in den nachfolgenden Zustand (alle Bits 0), wird hardwaremäßig das Timer-Flag TF 0 (1) gesetzt, welches von der Software ausgewertet und anschließend gelöscht werden kann.

Ändert die Interrupt-Routine nichts an den Zählerständen von TL 0 (1) und TH 0 (1), so wird im Timer-Mode nach jedem 98.304. (12 • 32 • 256) Oszillatortakt das Timer-Flag TF 0 (1) gesetzt. Damit läßt sich beispielsweise eine Interrupt-Routine aufrufen, in der die beiden Zähler mit neuen Werten geladen werden können, um eine Verkürzung der Timer-Durchlaufzeit zu erreichen.

Abbildung 97 zeigt ein Testprogramm, mit dessen Hilfe der Timer 0 in dem Mode 0 betrieben wird.

Die in dieser und in den nächsten Folgen vorgestellten Testprogramme sind in dem

EPROM ELV 9574 zusammengefaßt. Das Testprogramm ist für die Mikrocontroller-Grundlagenschaltung mit den angeschlossenen Schaltern bzw. 8-Bit-LED-Ausgabeplatinen vorgesehen. Die Schalterplatine ist im Gegensatz zu den bisherigen Testprogrammen an den Port P3 (ST 2) anzuschließen, während die 8-Bit-LED-Ausgabeplatine mit dem Port P1 über den Steckverbinder ST 3 mit der Grundschaltung zu verbinden ist.

Das Testprogramm 31 initialisiert zunächst den Timer 0 für die Betriebsart 0 (5-Bit-Vorteiler und 8-Bit-Teiler). Nach dem Start des Timers wird das Timer-Flag TF 0 112,5mal pro Sekunde (11,0592 MHz geteilt durch 12 • 32 • 256) gesetzt.

Die nachfolgende Schleife wartet, bis

das Timer-Flag gesetzt ist, löscht es anschließend, erhöht den Akkuinhalt um 1 und gibt dessen Inhalt am Port P1 aus, um die Anzeige über die 8 an P1 angeschlossenen Leuchtdioden vorzunehmen. Die höchstwertige LED D7 wird demnach alle 1,138 sek. ihren Zustand ändern, während die niederwertigen LEDs jeweils mit der doppelten Frequenz blinken.

## 4.2 Timer-Mode 1

Die Timer-Betriebsart 1 ist mit Ausnahme vom ersten Zähler TL 0 (1) identisch mit dem Mode 0. Im Mode 1 ist der Zähler TL 0 (1), wie auch der zweite 8-Bit-Zähler TH 0 (1) als 8-Bit-Zähler geschaltet. Mit diesem 16-Bit-Zähler läßt sich demnach ein maximaler Teilungsfaktor von 65.536 erreichen. Abbildung 98 zeigt zur Verdeutlichung die Funktionsweise des Timers in der Betriebsart 1.

Das in Abbildung 99 gezeigte Testprogramm 32 verhält sich in der grundlegenden Funktionsweise wie das Testprogramm 31 (Bild 97) mit dem Unterschied, daß der Timer 0 in der Betriebsart 1 läuft. Durch den 8-Bit-Vorteiler wird das Timer-Flag TF 0 nur ca. 14mal pro Sekunde (11,0592 MHz geteilt durch 12 • 256 • 256) gesetzt.




Bild 96: Funktionsweise des Timer/Zählers = (1) in der Betriebsart 0

```

025A 758900 L31:      MOV  TMOD, #00000000B ; Timer0: Gate aus,
                                ; Timer, Mode 0  Timer 1 keine Funktion
025D D28C            SETB  TR0 ; Timer0 starten
025F E4             CLR   A ; definierte Anfangsbedingung
0260 308DFDL31WDH:  JNB  TF0 L31WDH ; Warten, bis Timerflag gesetzt
0263 C28D            CLR   TF0 ; Timerflag löschen
0265 04             INC   A ; A = A +1
0266 F590            MOV  P1, A ; Ausgabe auf P1
0268 80F6            SJMP L31WDH ; Schleife
    
```

Bild 97: Testprogramm 31




Bild 98: Funktionsweise des Timers in der Betriebsart 1

Die Zustände der an P1 angeschlossenen Leuchtdioden werden sich also nur alle 70 ms ändern.

Das in Abbildung 100 dargestellte Testprogramm 33 entspricht vom Grundaufbau dem aus Abbildung 99 bekannten Pro-

gramm. Bei dieser Programmvariante wird zusätzlich die Gate-Steuerung eingeschaltet. Der durch 12 geteilte Oszillortakt kann nun nur auf die nachgeschalteten Zähler gelangen, wenn der Anschlußpin INT0 0 auf „aktiv-high“ liegt. Sobald der

dort angeschlossene Schalter S 2 an P3.2 den Low-Pegel legt, stoppt der Zähler. Anderenfalls wird der heruntergeteilte Oszillortakt auf den 16-Bit-Zähler geschaltet.

Die in Abbildung 101 und 102 gezeigten Testprogramme benutzen den Timer 0 des 8031/51 im Zähler-Mode. Das Taktsignal für die beiden nacheinander geschalteten 8-Bit-Zähler kommt direkt vom Port-Pin T 0 des Prozessors, wo der Schalter S 4 angeschlossen ist.

In der Hauptschleife wird nun der Zustand des ersten 8-Bit-Zählers zyklisch ausgelesen und am Port P 1 ausgegeben, an dem die LED-Ausgabeschaltung angeschlossen ist. Mit jedem Takt über S 4 erhöht sich der Inhalt des niederwertigen Zählers um 1, dessen aktueller Inhalt über D 0 bis D 7 angezeigt wird.

Bei dem in Abbildung 102 gezeigten Testprogramm 35 ist zusätzlich noch das Gate-Freigabebit gesetzt. Das an T 0 anliegende Taktsignal gelangt hier nur auf den nachgeschalteten Zähler, wenn am INT 0-Anschlußpin (S 2) High-Pegel anliegt. Bei anliegendem Low-Pegel ist der an T 0 anliegende Takt gesperrt.

```

026A 758901 L32:    MOV  TMOD,#00000001B ; Timer0: Gate aus,
                   ; Timer, Mode 1  Timer 1 keine Funktion
026D D28C          SETB TR0 ; Timer0 starten
026F E4           CLR  A ; definierte Anfangsbedingung
0270 308DFD L32WDH: JNB  TF0 L32WDH ; Warten, bis Timerflag gesetzt
0273 C28D          CLR  TF0 ; Timerflag löschen
0275 04           INC  A ; A = A +1
0276 F590          MOV  P1, A ; Ausgabe auf P1
0278 80F6          SJMP L32WDH ; Schleife
    
```

**Bild 99: Testprogramm 32**

```

027A 758909 L33:    MOV  TMOD,#00001001B ; Timer0: Gate aktiv,
                   ; Timer, Mode 1  Timer 1 keine Funktion
027D D28C          SETB TR0 ; Timer0 starten
027F E4           CLR  A ; definierte Anfangsbedingung
0280 308DFD L33WDH: JNB  TF0 L33WDH ; Warten, bis Timerflag gesetzt
0283 C28D          CLR  TF0 ; Timerflag löschen
0285 04           INC  A ; A = A +1
0286 F590          MOV  P1, A ; Ausgabe auf P1
0288 80F6          SJMP L33WDH ; Schleife
    
```

**Bild 100: Testprogramm 33**

```


028A 758905 L34:    MOV  TMOD,#00000101B ; Timer0: Gate aus,
                   ; Zähler, Mode 1  Timer 1 keine Funktion
028D 758A00        MOV  TL0,#000H ; Anfangszustand des Zählers
0290 D28C          SETB TR0 ; Timer0 starten
0292 E58A L34WDH:  MOV  A, TL0 ; Zählerstand lesen
0294 F590          MOV  P1, A ; Ausgabe auf P1
0296 80FA          SJMP L34WDH ; Schleife
    
```

**Bild 101: Testprogramm 34**

```

0298 75890D L35:    MOV  TMOD,#00001101B ; Timer0: Gate aktiv,
                   ; Zähler, Mode 1  Timer 1 keine Funktion
029B 758A00        MOV  TL0,#000H ; Anfangszustand des Zählers
029E D28C          SETB TR0 ; Timer0 starten
02A0 E58A L35WDH:  MOV  A, TL0 ; Zählerstand lesen
02A2 F590          MOV  P1, A ; Ausgabe auf P1
02A4 80FA          SJMP L35WDH ; Schleife
    
```

**Bild 102: Testprogramm 35**



**Bild 103: Funktion des Timers 0 (1) in der Betriebsart 2**

## 4.3 Timer-Mode 2

Abbildung 103 zeigt die Funktion des Timers 0 (1) in der Betriebsart 2. Dieser Mode wird am häufigsten benutzt, da ein vielfältiger Einsatz möglich ist. Der 8-Bit-Zähler TL 0 (1) wird wie in den anderen Betriebs-Modi auch über den externen Eingang oder den internen Oszillator getaktet. Der Überlauf des Zählers bewirkt, daß der in TH 0 (1) gespeicherte Zahlenwert in das Zählerregister kopiert wird, wobei der Inhalt von TH 0 (1) unverändert bleibt (Reload). Zusätzlich setzt jeder Überlauf das zugehörige Timerflag (TF 0 bzw. TF 1) und kann neben der Interruptauslösung auch als Schiebetakt für die serielle Stelle verwendet werden (nur Timer 1).

Dieser programmierbare Teilermode findet sehr oft Verwendung bei der Realisierung einer Echtzeituhr auf Sekunden-, Hundertstelsekunden- oder Tausendstelsekunden-Basis. Für den Betrieb der seriellen Schnittstelle wird dieser Betriebsmode vielfach für die Realisierung unterschiedlicher Baudraten (verschiedene Vorteiler) verwendet.

Abbildung 104 zeigt das Testprogramm 36, welches die Verwendung des Timer-Modus 2 demonstriert. Nach der Initialisierung und dem Starten des Timers wird in der äußeren Schleife der Reload-Wert vom Port P3 in den Reload-Speicher TH 0 geladen. Anschließend wird abgewartet, bis ein Timer-Überlauf erfolgt ist.

Nach Überlauf und Setzen des Timer-Flags durch den Prozessor wird dieses von

02A6 758902 L36:	MOV	TMOD, #0000010B	; Timer0: Gate aus, ; Timer, Mode 2	Timer 1 keine Funktion
02A9 7800	MOV	R0, #000H		; R0 Anfangszustand
02AB D28C	SETB	TR0		; Timer0 starten
02AD 85B08CL36WDH:	MOV	TH0, P3		; Reloadwert von P3 lesen
02B0 308DFDL36WDH1:	JNB	TF0 L36WDH1		; Warten, bis Timerflag gesetzt
02B3 C28D	CLR	TF0		; Timerflag löschen
02B5 D8F9	DJNZ	R0, L36WDH1		; Warten, bis R0 0 geworden ist
02B7 04	INC	A		; A = A + 1
02B8 F590	MOV	P1, A		; Ausgabe auf P1
02BA 80F1	SJMP	L36WDH		; Schleife

**Bild 104: Testprogramm 36**

der Software wieder gelöscht und mit dem folgenden DJNZ-Befehl die innere Schleife abgeschlossen. Nachdem diese 256mal durchlaufen ist, wird der Inhalt des Akkumulators um 1 erhöht und der aktuelle Zustand auf den Port P1 zur Aktivierung der angeschlossenen Leuchtdioden ausgegeben. Die äußere Schleife wiederholt permanent die Ausführung.

Je nach Einstellung der Schalter S 0 bis S 7 erfolgt das Erhöhen des Akkumulatorinhalts 14 bis 3600mal pro Sekunde. Im letztgenannten Fall läßt sich das Flackern der höchstwertigen Leuchtdiode (D 7) noch gerade erkennen.

#### 4.4 Timer-Mode 3

Eine Besonderheit stellt die Betriebsart 3 vom Timer 0 dar. Hier werden, wie in Abbildung 105 gezeigt, TL 0 und TH 0 als 2 getrennte 8-Bit-Zähler verwendet. Die Steuerung des TL 0-Zählers erfolgt wie in den anderen Betriebs-Modi auch über die Steuereinschübe C/T, TR 0 und Gate 0 sowie über die Prozessoranschlußpins T 0 und INT 0.

Ein Zählerüberlauf wird ebenfalls durch das Timer-Flag TF 0 angezeigt, woraufhin ein Timer-Interrupt ausgelöst werden kann.

Der Timer TH 0 benutzt in dieser Betriebsart das Steuerregister TR 1 sowie das Timer-Interrupt-Flag TF 1 und den zugehörigen Interrupt vom Timer-Zähler 1.

Der Zeitgeber 1 läßt sich dabei wie gewohnt in den Betriebsarten 0, 1 und 2 betreiben mit der Ausnahme, daß die vom Timer 0 belegten Flags (TR 1 und TF 1) nicht verwendbar sind. Daraus ergibt sich, daß die Ansteuerung des Timers 1 ausschließlich vom durch 12 geteilten Oszillatortakt erfolgen kann.

Das Ein- bzw. Ausschalten (Start/Stop) des Timers 1 erfolgt dadurch, daß die Betriebsart für diesen Timer auf 3 (M 0 und M 0 = 1) eingestellt wird.

Das Starten des Zählers wird durch die Umschaltung in die entsprechende Betriebsart vorgenommen. Die Zähler des Timers 1 lassen sich wie gewohnt beschreiben bzw. auslesen, wobei sich beim Überlauf von TH 1 kein Überlauf-Interrupt auslösen läßt, da TF 1 bereits, wie beschrieben, anderweitig verwendet wird. Der Schieberegister für die serielle Schnittstelle steht an diesem Ausgang für die Baudraten-Generierung aber weiterhin zur Verfügung.

Die Betriebsart 3 ist für solche Anwendungen vorgesehen, die einen weiteren 8-Bit-Zeitgeber oder Zähler benötigen. Damit verfügt der 8031/51-Prozessor über (wenn auch eingeschränkte) 3 voneinander unabhängige Timer bzw. Zähler.


Abbildung 106 zeigt das Testprogramm 37, welches die Verwendung des Timers 0 in der Betriebsart 3 demonstriert. Im Initialisierungsteil wird für den Timer 0 der Mode 3 eingestellt. Der Zähler TL 0 erhält seine Ansteuerung durch den Prozessoranschluß T 0, dessen Takt durch den Prozessoranschluß INT 0 freigegeben wird. Der zweite Zähler TH 0 wird ebenfalls freigegeben und fungiert als 8-Bit-Teiler. In der Hauptschleife wird der Inhalt des Zählers TL 0 gelesen und dessen niederwertige 4 Bits auf die Leuchtdioden D 0 bis D 3 an dem Treiberausgang von P1 ausgegeben.

Die folgende Schleifenstruktur, beginnend mit dem Befehl „JNB TF1 L37A“ erhöht mit jedem 256sten TH 0-Überlauf (TF 1) den Inhalt des Prozessorregisters R 2. Durch die anschließenden Befehle wird das niederwertige Nibble von R 2 über die Bits D 4 bis D 7 vom Port P 1 ausgegeben.

Die Leuchtdioden der 8-Bit-LED-Ausgabeschaltung D 4 bis D 7 ändern sich mit einer Frequenz von ca. 14 Hz, wobei sich der Inhalt von D 0 bis D 3 nur ändert, wenn der Schalter S 2 auf High-Pegel liegt und gleichzeitig am Prozessoranschluß T 0 (S 4) ein High-Low-Übergang stattfindet.

Der Timer 1 könnte nun mit den oben genannten Einschränkungen in den Betriebsarten 0 bis 2 verwendet werden.

Im nächsten Teil dieser Artikelserie beschreiben wir den Timer 2 des 8032/52 **ELV**



**Bild 105: Betriebsart 3 von Timer 0**

02BC 75893F L37:	MOV	TMOD, #00111111B	; Timer0: Gate aktiv, ; Zähler, Mode 3	Timer 1 gestoppt
02BF 7820	MOV	R0, #020H		; Zeiger initialisieren
02C1 7900	MOV	R1, #000H		; R1 Anfangszustand
02C3 7A00	MOV	R2, #000H		; R2 Anfangszustand
02C5 D28C	SETB	TR0		; Timer0 starten
02C7 D28E	SETB	TR1		; Timer1 starten
02C9 E58A L37WDH:	MOV	A, TL0		; Timer 0 lesen
02CB D6	XCHD	A, @R0		; Nur unteres Nibble tauschen
02CC 8690	MOV	P1, @R0		; Ausgabe auf P1
02CE 308F0B	JNB	TF1 L37A		; Springe, wenn Timerflag gesetzt
02D1 C28F	CLR	TF1		; Timerflag löschen
02D3 D907	DJNZ	R1, L37A		; Warten, bis R1 0 geworden ist
02D5 0A	INC	R2		; R2 = R2 + 1
02D6 EA	MOV	A, R2		; R2 -> A
02D7 C4	SWAP	A		; unteres und oberes Nibble tauschen
02D8 D6	XCHD	A, @R0		; Nur unteres Nibble tauschen
02D9 F6	MOV	@R0, A		; Wert speichern
02DA F590	MOV	P1, A		; Ausgabe auf P1
02DC 80EB L37A:	SJMP	L37WDH		; Schleife

**Bild 106: Testprogramm 37**